
Livex
    

This is not production code. It's a (fully functional) DX experiment to be able
to experiment with the combination of various features. But uses elixir trickery
and rather too much knowledge of liveview internals to make it work. Ideally the
end state is either for the features to be added to liveview, or for the
capability to add the features to be added.
Livex is a library that provides LiveView components and views with
enhanced state management, streamlining the persistence of state to the client
allowing automatic recovery of both url based and non url based client state.
It allows for more a more declarative style based on the render -> reducer ->
render cycle, being robust in the face of reconnections without needing to hand
persist state to the client via push_patch().

  
    
  
  Why Livex?


LiveView is a powerful platform for building interactive web applications.
However, as applications grow, certain patterns can become cumbersome:
	URL State Management: While storing state in the URL is crucial for
resilience (surviving refreshes, back button), manipulating individual query
parameters can be less ergonomic than direct state updates via assign().
	LiveComponent State: Managing state within LiveComponents robustly often
involves delegating to the parent and manually merging state concerns into the
URL query string, which can become complex.
	Clear State Patterns: Defining clear patterns for different types of state
(parent-owned, child-owned, persistent, ephemeral) and how they transition can
be challenging.
	Derived State Updates: Reacting to changes in URL parameters or component
arguments to refresh externally stored or derived state is often handled
imperatively, whereas a more declarative, event-driven approach could be
beneficial.
	Component PubSub: Enabling components to be self-sufficient in terms of
real-time messaging via PubSub can require boilerplate.

Livex aims to address these points by introducing features that promote a more
declarative and streamlined developer experience, fitting within the existing
LiveView paradigm while borrowing well-established concepts. It's intended
to be a proof of concept of features that would be natural evolutions of
LiveView and hopefully may lead to such features being included.

  
    
  
  Installation


The package can be installed by adding livex to your list of dependencies in
mix.exs:
def deps do
  [
    {:livex, "~> 0.2.0"}
  ]
end

  
    
  
  Demo


To see Livex in action with concrete examples, check out the demo application at
https://github.com/u2i/livex_demo.

  
    
  
  Features



  
    
  
  Declarative State Definition (state)


Livex allows you to declaratively define state properties for your LivexView or
LivexComponent.
defmodule MyAppWeb.ArticleView do
  use MyAppWeb, :livex_view # Or use Livex.LivexView directly

  # State stored in URL, survives refreshes, typed as :integer
  state :page_number, :integer, url?: true
  # State stored in URL, survives refreshes, typed as :string
  state :category_filter, :string, url?: true
  # Client-side state, survives reconnects but not refreshes, typed as :boolean
  state :show_advanced_search, :boolean

  # Example of a component state type.
  # The `MyAppWeb.Components.CommentForm` component's attrs will be managed here.
  # This allows the parent view to control the initial state or reset the form.
  state :comment_form, MyAppWeb.Components.CommentForm
end

defmodule MyAppWeb.Components.InteractiveCard do
  use MyAppWeb, :livex_component # Or use Livex.LivexComponent directly

  # State specific to this card instance, can be stored in URL if needed
  state :display_mode, :atom # :summary or :detailed
  state :user_rating, :integer

  # rest of your component code...
end
	state <param_name>, <type>: Defines a piece of state.
	url?: true: Indicates the state should be stored in the URL query string.
This makes the state bookmarkable and persistent across page refreshes.
	url?: false (or omitted): Indicates the state is stored only on the
client-side LiveView state. It will survive client-server reconnects but not
full page refreshes.
	Type System: Livex handles casting the state from the URL (string) or
client-side representation back to its defined Elixir type.
	Component State: You can define a state property as a component module
(e.g., state :new_comment_form, MyAppWeb.Components.CommentForm). Livex will
then manage the attributes (attr) of this component as part of the parent's
state. When this state is updated (e.g., via JSX.assign_state or a direct
assign in an event), Livex facilitates the re-rendering of the child component
with the new attribute values.
	Initial Values: Default values are not directly supported in the state
macro (yet). To set an initial value, use assign_new/2 within the pre_render/1
callback without dependencies, e.g., assign_new(socket, :my_state, fn -> initial_value end).


  
    
  
  Declarative Data Derivation and Lifecycle Management (pre_render, assign_new, stream_new)


Livex aims to simplify the LiveView/LiveComponent lifecycle (mount, update,
handle_params) by consolidating data derivation and initial assignment logic
into a pre_render/1 callback, coupled with enhanced assign and stream functions.
The typical flow becomes: render (pre_render, render)-> event -> reducer
(handle_event, handle_info, handle_async) -> render etc.
The pre_render callback largely replaces both mount and handle_params, providing
a single place to handle state initialization and data derivation. The standard
LiveView callbacks still exist and can be used when needed, but pre_render
provides a more declarative approach.
Example:
defmodule MyAppWeb.ProductListingView do
  use MyAppWeb, :livex_view

  state :selected_category, :string, url?: true
  state :sort_by, :atom, url?: true
  state :modal_open, :boolean

  def pre_render(socket) do
    {:noreply,
     socket
     |> assign_new(:modal_open, fn -> false end)
     |> assign_new(:page_title, fn -> "Product Catalog" end)
     |> assign_new(:selected_category, fn -> "all" end)
     |> assign_new(:sort_by, fn -> :name_asc end)
     |> stream_new(:products, [:selected_category, :sort_by], fn assigns ->
       # This function re-runs if selected_category or sort_by changes
       Products.list_available_products(
         category: assigns.selected_category,
         order_by: assigns.sort_by
       )
     end)}
  end

  # ... rest of the view code
end
	assign_new/4: Works like Phoenix.Component.assign_new/3,
but with an explicit list of dependencies (assign keys). The assignment
function will only re-run if one of these dependencies has changed. This
memoization helps avoid redundant computations or data fetching. Can also be
used without dependencies to set initial values.

Example:
defmodule MyAppWeb.UserProfileView do
  use MyAppWeb, :livex_view

  state :user_id, :string, url?: true # e.g., from URL like /users/:user_id

  def pre_render(socket) do
    {:noreply,
     socket
     |> assign_new(:profile, [:user_id], fn assigns ->
       # This function only runs if assigns.user_id changes
       Accounts.get_user_profile(assigns.user_id)
     end)
     |> assign_new(:activity_feed, [:user_id], fn assigns ->
       Activity.fetch_feed_for_user(assigns.user_id)
     end)
     |> assign_new(:is_admin_view, fn -> false end) # Initial assignment
    }
  end
  # ... render, handle_event, etc.
end
If :user_id (which is a state field) changes, the functions to fetch :profile
and :activity_feed will be re-executed.
	stream_new/4: An enhancement to LiveView's stream/4. It takes an
additional list of dependencies (assign keys from state or attr). The stream
will be reset and re-populated using the provided function if any of these
dependencies change. This is particularly useful for streams whose contents
depend on filter parameters or other dynamic data.

Example:
def pre_render(socket) do
  {:noreply,
   socket
   # Set initial values for filters if not present in URL
   |> assign_new(:filter_category, fn -> nil end)
   |> assign_new(:filter_price_range, fn -> nil end)
   # Stream depends on the potentially initialized state values
   |> stream_new(:products, [:filter_category, :filter_price_range], fn assigns ->
     filter_products(
       Products.list_all(),
       assigns.filter_category,
       assigns.filter_price_range
     )
   end)}
end
If either :filter_category or :filter_price_range changes, the :products stream will be
automatically updated by re-invoking the stream configuration function with the
new assigns.

  
    
  
  Component Properties (attr)


Used in LivexComponent to define properties passed down from a parent. These are
analogous to "props" in React and are fully controlled by the parent.
# In a filter component
attr :id, :string
attr :selected_value, :string
attr :options, :list
attr :title, :string
Components can use pre_render to initialize state based on attr values:
def pre_render(socket) do
  {:noreply,
   socket
   |> assign_new(:is_open, fn -> false end)
   |> assign_new(:pending_selection, [:selected_value], &(&1.selected_value))
   |> assign_new(:has_changes, fn -> false end)
   |> then(fn socket ->
     # Compute derived state
     assign(socket, :has_changes, socket.assigns.pending_selection != socket.assigns.selected_value)
   end)}
end
Component Modules as State Types
One of the powerful features of Livex is the ability to define a state property
using a component module as its type. This creates a tight integration between
parent views and their child components.
When you define a state property with a component module as its type:
# In parent view
state :edit_form, MyAppWeb.Components.EditForm
Livex will:
	Automatically manage the component's attributes as part of the parent's state
	Allow the parent to control the component's initial state or reset it
	Ensure proper re-rendering when the component state changes

The component itself defines its expected attributes using attr:
# In MyAppWeb.Components.EditForm
attr :id, :string, required: true
attr :item_id, :string
attr :action, :atom, default: :new
attr :on_save, :any
When the parent renders this component, it can pass the state directly:
<.live_component
  module={MyAppWeb.Components.EditForm}
  id="edit-form"
  {@edit_form}
/>
This pattern is especially useful for modal forms and other complex UI elements
that need to be controlled by a parent view. The parent can:
	Initialize the component state: assign(socket, :edit_form, %{item_id: "123", action: :edit})
	Reset the component: assign(socket, :edit_form, nil)
	Update specific attributes: update(socket, :edit_form, &Map.put(&1, :action, :new))

All while maintaining proper type safety and state persistence according to the
component's defined attributes.

  
    
  
  Simplified State Updates (JSX.assign_state)


For components, Livex introduces JSX.assign_state (used within HEEx templates) as
a convenient way to update its own state properties. This is analogous to the
setter functions returned by useState in React functional components. It
simplifies common cases of updating component state directly from template
events without needing a full handle_event callback for simple assignments.
Example:
defmodule MyAppWeb.Components.CounterButton do
  use MyAppWeb, :livex_component

  state :current_count, :integer
  state :is_highlighted, :boolean

  def pre_render(socket) do
    # Set initial values using assign_new without dependencies
    {:noreply,
      socket
      |> assign_new(:current_count, fn -> 0 end)
      |> assign_new(:is_highlighted, fn -> false end)
    }
  end

  def render(assigns) do
    ~H"""
    <div>
      <p class={if @is_highlighted, do: "font-bold", else: ""}>
        Count: {@current_count}
      </p>
      <button phx-click={JSX.assign_state(:current_count, @current_count + 1)}>
        Increment
      </button>
      <button phx-click={JSX.assign_state(is_highlighted: !@is_highlighted)}>
        Toggle Highlight
      </button>
      <button phx-click={JSX.assign_state(current_count: 0, is_highlighted: false)}>
        Reset
      </button>
    </div>
    """
  end
end
Clicking these buttons will directly update the :current_count and/or
:is_highlighted state of the CounterButton component instance, triggering a
re-render. JSX.assign_state can update one or multiple state fields. For more
complex state transitions or side effects, you would still use handle_event.
Example of conditional state updates:
<button
  type="button"
  phx-click={
    if @is_expanded do
      JSX.assign_state(
        is_expanded: false,
        pending_value: @initial_value
      )
    else
      JSX.assign_state(is_expanded: true)
    end
  }
>
  {if @is_expanded, do: "Close", else: "Expand"}
</button>

  
    
  
  Component Events (push_emit, phx-event)


Livex enhances event handling for components, allowing them to emit custom
events that parent LiveViews or LiveComponents can listen for, similar to how
standard phx-click and other DOM events work.
	Emitting from the component's template: Use JSX.emit(:event_name, value:
%{...}) within a phx-click or other event binding.
	Emitting from the component's Elixir code: Use push_emit(socket,
:eventname, payload \ %{}). I'm considering removing this, the use cases
for this may be better handled by send_message.

Example:
Child Component (MyAppWeb.Components.EditableField):
defmodule MyAppWeb.Components.EditableField do
  use MyAppWeb, :livex_component

  attr :field_id, :string, required: true
  attr :initial_value, :string
  state :current_value, :string
  state :is_editing, :boolean

  def pre_render(socket) do
    # Initialize current_value from attr and set editing to false initially
    {:noreply,
      socket
      |> assign_new(:current_value, [:initial_value], &(&1.initial_value))
      |> assign_new(:is_editing, fn -> false end)
    }
  end

  def render(assigns) do
    ~H"""
    <div :if={@is_editing}>
      <input type="text" phx-target={@myself} phx-change="update_value" value={@current_value} />
      <button phx-target={@myself} phx-click="save_changes">Save</button>
      <button phx-target={@myself} phx-click={JSX.assign_state(is_editing: false)}>Cancel</button>
    </div>
    <div :if={!@is_editing}></div>
      <span>{@current_value}</span>
      <button phx-target={@myself} phx-click={JSX.assign_state(is_editing: true)}>Edit</button>
    </div>
    """
  end

  def handle_event("update_value", %{"value" => new_val}, socket) do
    {:noreply, assign(socket, :current_value, new_val)}
  end

  def handle_event("save_changes", _, socket) do
    # Emit an event to the parent with the field_id and new value
    socket = push_emit(socket, :field_updated, %{id: socket.assigns.field_id, value: socket.assigns.current_value})
    {:noreply, assign(socket, :is_editing, false)}
  end
end
Parent View (MyAppWeb.SettingsView):
defmodule MyAppWeb.SettingsView do
  use MyAppWeb, :livex_view

  state :username, :string
  state :email, :string

  def pre_render(socket) do
    # Set initial values for settings
    {:noreply,
      socket
      |> assign_new(:username, fn -> "User123" end)
      |> assign_new(:email, fn -> "user@example.com" end)
    }
  end

  def render(assigns) do
    ~H"""
    <div>
      <h2>Settings</h2>
      <label>Username:</label>
      <.live_component
        module={MyAppWeb.Components.EditableField}
        id={"username_field"}
        field_id="username"
        initial_value={@username}
        phx-field_updated="handle_setting_change"
      />
      <label>Email:</label>
      <.live_component
        module={MyAppWeb.Components.EditableField}
        id={"email_field"}
        field_id="email"
        initial_value={@email}
        phx-field_updated="handle_setting_change"
      />
    </div>
    """
  end

  def handle_event("handle_setting_change", %{"id" => "username", "value" => new_username}, socket) do
    # Persist username change, update local state
    {:noreply, assign(socket, :username, new_username)}
  end

  def handle_event("handle_setting_change", %{"id" => "email", "value" => new_email}, socket) do
    # Persist email change, update local state
    {:noreply, assign(socket, :email, new_email)}
  end
end
Internal Messaging: send_message & handle_message
The DOM messages have their limitations and are best used to be able to
respond to DOM interactions. Indeed the LiveView documentation describes a
mechanism for supporting standardized hooks between child -> parent.
But we can make this more ergonomic. So Livex provides a lightweight internal
messaging mechanism through send_message and handle_message.
Key differences from DOM events:
	send_message is called from handle_* functions, not from templates
	Messages are handled by handle_message callbacks, not handle_event
	Primarily designed for child-to-parent component communication initiated
from the server. However, you can specify any component or the top level
view as the target.
	Provides a cleaner separation between UI events and internal component logic
making for a clean external interface.
	Supports pattern matching on source module name: to avoid issues with
conflicting event names

Example of child-to-parent communication:
# Child Component
defmodule MyAppWeb.FilterComponent do
  use MyAppWeb, :livex_component

  attr :id, :string, required: true
  attr :available_filters, :list, required: true
  state :selected_filter, :string

  def handle_event("select_filter", %{"filter" => filter}, socket) do
    # Update local state
    socket = assign(socket, :selected_filter, filter)

    # Notify parent about the filter change
    socket = send_message(socket, {:filter_changed, filter})

    {:noreply, socket}
  end

  def render(assigns) do
    ~H"""
    <div class="filter-panel">
      <h3>Filters</h3>
      <div :for={filter <- @available_filters} class="filter-options">
        <button
          class={if @selected_filter == filter, do: "selected", else: ""}
          phx-target={@myself}
          phx-click="select_filter"
          phx-value-filter={filter}
        >
          <%= filter %>
        </button>
      </div>
    </div>
    """
  end
end

# Parent LiveView
defmodule MyAppWeb.ProductsView do
  use MyAppWeb, :livex_view

  state :products, :list
  state :current_filter, :string

  def pre_render(socket) do
    {:noreply,
     socket
     |> assign_new(:products, fn -> fetch_products() end)
     |> assign_new(:current_filter, fn -> "all" end)}
  end

  # Handle the message sent from the child component
  def handle_message(FilterComponent, :filter_changed, filter, socket) do
    # Update the product list based on the new filter
    filtered_products = filter_products(socket.assigns.products, filter)

    {:noreply,
     socket
     |> assign(:current_filter, filter)
     |> assign(:products, filtered_products)}
  end

  def render(assigns) do
    ~H"""
    <div class="products-page">
      <h1>Products ({@current_filter})</h1>

      <.live_component
        module={MyAppWeb.FilterComponent}
        id="product-filter"
        target={@myself}
        available_filters={["all", "electronics", "clothing", "books"]}
      />

      <div :for={product <- @products} class="product-list">
        <div class="product-card">
          <h3>{product.name}</h3>
          <p>{product.price}</p>
        </div>
      </div>
    </div>
    """
  end

  defp filter_products(products, "all"), do: products
  defp filter_products(products, category) do
    Enum.filter(products, &(&1.category == category))
  end

  defp fetch_products do
    # Fetch products from database or API
    [
      %{name: "Laptop", price: "$999", category: "electronics"},
      %{name: "T-shirt", price: "$25", category: "clothing"},
      %{name: "Novel", price: "$15", category: "books"}
    ]
  end
end
When to use send_message/handle_message:
	For child-to-parent component communication and related activities
	When you want to separate UI event handling from business logic
	To create a cleaner separation of concerns in complex component hierarchies
	When you need more direct communication than what phx-<event> provides


  
    
  
  Enable PubSub for Components (assign_topic)


 Not Implemented Yet! 
Livex aims to make it easier for components to subscribe to PubSub topics
themselves, especially when the topic depends on the component's attrs or state.
assign_topic/4 works similarly to assign_new/4, subscribing the component to a
PubSub topic. The topic name can be dynamically generated based on dependencies.
If those dependencies change, the component would be re-subscribed to the new
topic.
defmodule MyAppWeb.Components.RealtimeDocumentStatus do
  use MyAppWeb, :livex_component

  attr :document_id, :uuid, required: true
  state :status_message, :string

  def pre_render(socket) do
    {:noreply,
     socket
     # Set initial status
     |> assign_new(:status_message, fn -> "Connecting..." end)
     # Subscribe to topic based on attr
     |> assign_topic(:doc_status_updates, [:document_id], fn assigns ->
       # Dynamically generate topic based on document_id
       "document_updates:#{assigns.document_id}"
     end)}
  end

  # This will be called when a message is published to the subscribed topic
  def handle_info({:doc_status_updates, %{message: msg}}, socket) do
    # Update component state based on the PubSub message
    {:noreply, assign(socket, :status_message, msg)}
  end

  def render(assigns) do
    ~H"""
    <div class="status-badge">
      Document {@document_id}: {@status_message}
    </div>
    """
  end
end

  
    
  
  Summary


By combining these features, Livex aims to provide:
	A cleaner conceptual model for LiveView and LiveComponent state and lifecycle.
	Reduced boilerplate for common patterns like URL state management, derived
data, and component communication.
	More declarative data flow, where changes to state or attrs automatically
propagate to derived data and streams via pre_render and dependency-aware
functions like assign_new and stream_new.
	Avoidance of many manual push_patch calls, as state changes naturally lead to
re-renders.


  
    
  
  Contributing


Contributions are welcome! This is an experimental project, so feel free to open
issues or submit pull requests with ideas or improvements.

  
    
  
  License


This project is licensed under the MIT License - see the LICENSE file for
details.

  
    
  
  Setup Instructions


	Add Livex to your dependencies

Add livex to your dependencies in mix.exs:
def deps do
  [
    {:livex, "~> 0.1.0"}
  ]
end
	Configure your web module

Update your web module (typically lib/my_app_web.ex) to include Livex view and
component definitions. This usually involves defining livex_view/0 and
livex_component/0 functions that use Livex.LivexView or Livex.LivexComponent
respectively, along with Livex.JSX and any shared HTML helpers.
Example lib/my_app_web.ex:
defmodule MyAppWeb do
  # ... other functions like static_paths, router, channel, controller ...

  def livex_view do
    quote do
      use Phoenix.LiveView,
        layout: {MyAppWeb.Layouts, :app} # Or your desired layout

      unquote(view_helpers()) # Keep your existing view_helpers

      # Livex specific uses
      use Livex.LivexView
      use Livex.JSX # For JSX.emit, JSX.assign_state etc.
    end
  end

  def livex_component do
    quote do
      use Phoenix.LiveComponent

      # Livex specific uses
      use Livex.LivexComponent
      use Livex.JSX # For JSX.emit, JSX.assign_state etc.
    end
  end

  def view_helpers do
    quote do
      use Phoenix.HTML
      import Phoenix.LiveView.Helpers
      import Phoenix.View
      # ... other helpers like ErrorHelpers, Gettext, Routes, CoreComponents
      # import MyAppWeb.CoreComponents # If you have them
    end
  end

  # ... (rest of your web module, e.g., verified_routes)
end
	JavaScript Integration

Add these two lines to your assets/js/app.js file:
// Add this import near your other imports (after LiveSocket import)
import { enhanceLiveSocket } from "livex";

// Add this line after creating your LiveSocket instance but before connecting
// Example:
// let csrfToken = document.querySelector("meta[name='csrf-token']").getAttribute("content")
// let liveSocket = new LiveSocket("/live", Socket, {params: {_csrf_token: csrfToken}})
liveSocket = enhanceLiveSocket(liveSocket); // Add this line
// liveSocket.connect()
	Using Livex in your application

Create LiveView modules using the new livex_view definition from your web module:
defmodule MyAppWeb.MyPageView do
  use MyAppWeb, :livex_view # Assuming :livex_view is defined in MyAppWeb

  state :item_id, :string, url?: true
  state :is_editing_mode, :boolean

  # Your view code...
end
Create LiveComponents using the new livex_component definition:
defmodule MyAppWeb.MyItemComponent do
  use MyAppWeb, :livex_component # Assuming :livex_component is defined in MyAppWeb

  attr :item_name, :string, required: true
  state :show_details, :boolean

  # Your component code...
end



  

    
Livex.Handlers 
    



      
Handlers for component events in Livex.
This module contains functions for handling events that are dispatched to components.

      


      
        Summary


  
    Functions
  


    
      
        handle_component_event(module, params, socket)

      


        Handles component events by mapping parameters and calling pre_render.



    





      


      
        Functions

        


  
    
      
    
    
      handle_component_event(module, params, socket)



    

  


  

Handles component events by mapping parameters and calling pre_render.
This function is used internally by Livex to process events that are sent to components.

  
    
  
  Parameters


	module - The component module
	params - The event parameters
	socket - The LiveView socket


  
    
  
  Returns


	{:noreply, socket} - The updated socket after processing the event


  


        

      


  

    
Livex.JSX 
    



      
Provides macros for client-side state updates and event emission in Livex components.
This module contains two main sets of functionality:
	JSX.emit - Emit events from components to parent views/components
	JSX.assign_state - Update component state directly from templates


  
    
  
  Event Emission (JSX.emit)


The emit macros allow components to emit events that can be handled by parent components
or views. This creates a clean communication channel between child and parent components.

  
    
  
  Examples


# In a component template
<button phx-click={JSX.emit(:close)}>Cancel</button>
<button phx-click={JSX.emit(:save, value: %{id: @id, data: @form_data})}>Save</button>

# In the parent component/view
<.live_component 
  module={MyApp.ModalComponent} 
  id="my-modal" 
  phx-close="handle_modal_close"
  phx-save="handle_modal_save">
  Modal content here
</.live_component>

# In the parent's handle_event
def handle_event("handle_modal_close", _, socket) do
  {:noreply, assign(socket, :modal_open, false)}
end

def handle_event("handle_modal_save", %{"id" => id, "data" => data}, socket) do
  # Process the saved data
  {:noreply, socket}
end

  
    
  
  State Updates (JSX.assign_state)


The assign_state macros allow updating component state directly from templates,
without needing to write handle_event callbacks for simple state changes.

  
    
  
  Examples


# Update a single value
<button phx-click={JSX.assign_state(:is_expanded, true)}>Expand</button>

# Update multiple values at once
<button phx-click={JSX.assign_state(is_expanded: false, selected_tab: "details")}>
  Close
</button>

# Conditional updates
<button phx-click={
  if @is_expanded do
    JSX.assign_state(is_expanded: false, pending_value: @initial_value)
  else
    JSX.assign_state(is_expanded: true)
  end
}>
  {if @is_expanded, do: "Close", else: "Expand"}
</button>

# Force a component refresh without changing state
<button phx-click={JSX.assign_state()}>Refresh Component</button>

  
    
  
  Usage


To use these macros, add use Livex.JSX to your module, or they are automatically
imported when using Livex.LivexView or Livex.LivexComponent.

      


      
        Summary


  
    Functions
  


    
      
        assign_state()

      


        Creates a JS command to trigger a component update without changing any data.



    


    
      
        assign_state(opts)

      


        Creates a JS command to update component data from the client with a keyword list of options.



    


    
      
        assign_state(key, val)

      


        Creates a JS command to update a single component data value from the client.



    


    
      
        emit(event_name_suffix_ast)

      


    


    
      
        emit(js_ast, event_name_suffix_ast)

      


    


    
      
        emit(js_ast, event_name_suffix_ast, opts_list_ast)

      


        Generates a JS command to emit a client-side event.



    





      


      
        Functions

        


  
    
      
    
    
      assign_state()


        (macro)


    

  


  

Creates a JS command to trigger a component update without changing any data.
This is useful for forcing a component to re-render.

  
    
  
  Examples


<button phx-click={JSX.assign_state()}>Refresh Component</button>

  



  
    
      
    
    
      assign_state(opts)


        (macro)


    

  


  

Creates a JS command to update component data from the client with a keyword list of options.
This macro allows updating multiple component state values directly from client-side events.

  
    
  
  Parameters


	opts - A keyword list of key-value pairs to update in the component's state


  
    
  
  Examples


<button phx-click={JSX.assign_state(is_expanded: true, selected_tab: "details")}>
  Expand Details
</button>

  



  
    
      
    
    
      assign_state(key, val)


        (macro)


    

  


  

Creates a JS command to update a single component data value from the client.
See assign_state/1 for updating multiple values at once.

  
    
  
  Parameters


	key - The key to update in the component's state
	val - The value to assign


  
    
  
  Examples


<button phx-click={JSX.assign_state(:is_expanded, true)}>Expand</button>

  



  
    
      
    
    
      emit(event_name_suffix_ast)


        (macro)


    

  


  


  



  
    
      
    
    
      emit(js_ast, event_name_suffix_ast)


        (macro)


    

  


  


  



  
    
      
    
    
      emit(js_ast, event_name_suffix_ast, opts_list_ast)


        (macro)


    

  


  

Generates a JS command to emit a client-side event.

  
    
  
  Variants


	emit(js, event_suffix, opts) - Chain on an existing JS struct with options
	emit(js, event_suffix) - Chain on an existing JS struct without options
	emit(event_suffix, opts) - Create a new JS chain with options
	emit(event_suffix) - Create a new JS chain without options


  
    
  
  Parameters


	js - An existing Phoenix.LiveView.JS struct to chain operations on
	event_suffix - The suffix of the event to emit (e.g., "close" for "phx-close")
	opts - Options for the event, such as values: %{...}


  
    
  
  Examples


# With existing JS chain and options
JS.transition(...) |> JSX.emit("close", values: %{reason: "cancelled"})

# With existing JS chain, no options
JS.transition(...) |> JSX.emit("close")

# New JS chain with options
JSX.emit("close", values: %{reason: "cancelled"})

# New JS chain without options
JSX.emit("close")

  


        

      


  

    
Livex.LivexComponent 
    



      
A module that enhances Phoenix LiveComponent with automatic state management and lifecycle improvements.

  
    
  
  Features


	Declarative State Definition: Define component state properties with type safety
	Component Properties (attr): Define expected properties passed from parent
	Simplified Lifecycle: Consolidate data derivation logic in the pre_render callback
	Dependency-Aware Assignments: Use assign_new/4 to compute values only when dependencies change
	Simplified State Updates: Use JSX.assign_state for direct state updates from templates
	Component Events: Emit events to parent views with push_emit or JSX.emit
	PubSub Integration: Subscribe to topics with dependency-aware assign_topic


  
    
  
  State and Properties


Livex components use two key concepts for managing data:
	state - Internal component state that persists across renders
	attr - Properties passed down from parent components/views


  
    
  
  Component State


# Client-side state, survives reconnects but not refreshes
state :is_expanded, :boolean
state :selected_tab, :string
state :current_count, :integer

  
    
  
  Component Properties (attr)


# Properties passed from parent
attr :id, :string, required: true
attr :item_id, :string
attr :initial_value, :string
attr :on_save, :any

  
    
  
  Lifecycle Management


Livex simplifies the component lifecycle by consolidating data derivation and initial 
assignment logic into a pre_render/1 callback. This is where you can initialize state 
based on attr values or compute derived data.
def pre_render(socket) do
  {:noreply,
   socket
   |> assign_new(:is_expanded, fn -> false end)
   |> assign_new(:pending_selection, [:selected_value], &(&1.selected_value))
   |> assign_new(:has_changes, fn -> false end)
   |> then(fn socket ->
     # Compute derived state
     assign(socket, :has_changes, 
       socket.assigns.pending_selection != socket.assigns.selected_value)
   end)}
end

  
    
  
  Simplified State Updates


Livex introduces JSX.assign_state for updating component state directly from templates:
# Update a single value
<button phx-click={JSX.assign_state(:is_expanded, true)}>Expand</button>

# Update multiple values
<button phx-click={JSX.assign_state(is_expanded: false, selected_tab: "details")}>
  Close
</button>

# Conditional updates
<button phx-click={
  if @is_expanded do
    JSX.assign_state(is_expanded: false, pending_value: @initial_value)
  else
    JSX.assign_state(is_expanded: true)
  end
}>
  {if @is_expanded, do: "Close", else: "Expand"}
</button>

  
    
  
  Component Events


Livex enhances event handling for components, allowing them to emit custom events 
that parent LiveViews or LiveComponents can listen for:

  
    
  
  Emitting from templates:


<button phx-click={JSX.emit(:save, value: %{id: @id, data: @form_data})}>
  Save Changes
</button>

  
    
  
  Emitting from Elixir code:


def handle_event("save_changes", _, socket) do
  # Process the data...
  
  # Emit an event to the parent
  socket = push_emit(socket, :saved, %{id: socket.assigns.id})
  {:noreply, assign(socket, :is_saving, false)}
end

  
    
  
  Handling in parent:


<.live_component
  module={MyApp.FormComponent}
  id="my-form"
  phx-saved="handle_form_saved"
/>

# In the parent's handle_event
def handle_event("handle_form_saved", %{"id" => id}, socket) do
  # Handle the saved event
  {:noreply, socket}
end

  
    
  
  PubSub Integration


Livex makes it easier for components to subscribe to PubSub topics with assign_topic:
def pre_render(socket) do
  {:noreply,
   socket
   |> assign_new(:status_message, fn -> "Connecting..." end)
   |> assign_topic(:doc_updates, [:document_id], fn assigns ->
     "document_updates:#{assigns.document_id}"
   end)}
end

# Handle PubSub messages
def handle_info({:doc_updates, %{message: msg}}, socket) do
  {:noreply, assign(socket, :status_message, msg)}
end

  
    
  
  Complete Example


defmodule MyApp.FormComponent do
  use Livex.LivexComponent
  
  attr :id, :string, required: true
  attr :item_id, :string
  attr :on_save, :any
  
  state :is_expanded, :boolean
  state :form_data, :map
  state :is_valid, :boolean
  
  def pre_render(socket) do
    {:noreply,
     socket
     |> assign_new(:is_expanded, fn -> false end)
     |> assign_new(:form_data, [:item_id], fn assigns ->
       if assigns.item_id, do: load_item(assigns.item_id), else: %{}
     end)
     |> assign_new(:is_valid, [:form_data], fn assigns ->
       validate_form(assigns.form_data)
     end)}
  end
  
  def render(assigns) do
    ~H"""
    <div id={@id} class="form-component">
      <h3>Edit Item</h3>
      
      <div class="form-fields">
        <!-- Form fields here -->
      </div>
      
      <div class="actions">
        <button 
          disabled={!@is_valid}
          phx-click="save"
          phx-target={@myself}>
          Save
        </button>
        
        <button phx-click={JSX.emit(:cancel)}>
          Cancel
        </button>
      </div>
    </div>
    """
  end
  
  def handle_event("save", _, socket) do
    # Process the save...
    if socket.assigns.on_save do
      socket.assigns.on_save.(socket.assigns.form_data)
    end
    
    # Emit an event to the parent
    socket = push_emit(socket, :saved, %{id: socket.assigns.id})
    {:noreply, socket}
  end
  
  defp load_item(id) do
    # Load item data
  end
  
  defp validate_form(data) do
    # Validate form data
  end
end

      


      
        Summary


  
    Functions
  


    
      
        inject_after_div(rendered, module, assigns)

      


    


    
      
        override_update(assigns, socket, module, _)

      


    


    
      
        override_update(assigns, socket, current_params, module, super)

      


    





      


      
        Functions

        


  
    
      
    
    
      inject_after_div(rendered, module, assigns)



    

  


  


  



  
    
      
    
    
      override_update(assigns, socket, module, _)



    

  


  


  



  
    
      
    
    
      override_update(assigns, socket, current_params, module, super)



    

  


  


  


        

      


  

    
Livex.LivexView 
    



      
A module that enhances Phoenix LiveView with automatic state management and lifecycle improvements.

  
    
  
  Features


	Declarative State Definition: Define state properties with type safety and URL persistence options
	Automatic URL State Management: State marked with url?: true is automatically persisted to the URL
	Simplified Lifecycle: Consolidate data derivation logic in the pre_render callback
	Dependency-Aware Assignments: Use assign_new/4 to compute values only when dependencies change
	Reactive Data Streams: Use stream_new/4 to automatically update streams when dependencies change
	Component State Management: Define component modules as state types for parent-controlled components


  
    
  
  State Management


Livex allows you to declaratively define state properties for your LiveView:
# State stored in URL, survives refreshes, typed as :integer
state :page_number, :integer, url?: true

# State stored in URL, survives refreshes, typed as :string
state :category_filter, :string, url?: true

# Client-side state, survives reconnects but not refreshes, typed as :boolean
state :show_advanced_search, :boolean

# Component state type - parent can control this component's attributes
state :edit_form, MyApp.Components.EditForm
	url?: true - State is stored in the URL query string, making it bookmarkable and persistent across page refreshes
	url?: false (or omitted) - State is stored only in client-side LiveView state, surviving reconnects but not refreshes
	Type System - Livex handles casting state from URL (string) or client-side representation to its defined Elixir type
	Component State - When a component module is used as a type, the parent can control that component's attributes


  
    
  
  Lifecycle Management


Livex simplifies the LiveView lifecycle by consolidating data derivation and initial assignment logic 
into a pre_render/1 callback. This largely replaces both mount/3 and handle_params/3, providing 
a single place to handle state initialization and data derivation.
The typical flow becomes: render -> event -> reducer (handle_event, handle_info, handle_async) -> pre_render -> render.
def pre_render(socket) do
  {:noreply,
   socket
   |> assign_new(:page_title, fn -> "Product Catalog" end)
   |> assign_new(:selected_category, fn -> "all" end)
   |> assign_new(:products, [:selected_category], fn assigns ->
     # This function only runs if selected_category changes
     Products.list_by_category(assigns.selected_category)
   end)}
end

  
    
  
  Dependency-Aware Assignments


The assign_new/4 function extends Phoenix.Component.assign_new/3 with dependency tracking:
# Runs only when user_id changes
assign_new(socket, :profile, [:user_id], fn assigns ->
  Accounts.get_user_profile(assigns.user_id)
end)

# Initial assignment (no dependencies)
assign_new(socket, :is_admin_view, fn -> false end)

  
    
  
  Reactive Data Streams


The stream_new/4 function enhances LiveView's stream/4 with dependency tracking:
# Stream resets and repopulates when filter_category or filter_price_range changes
stream_new(socket, :products, [:filter_category, :filter_price_range], fn assigns ->
  filter_products(
    Products.list_all(),
    assigns.filter_category,
    assigns.filter_price_range
  )
end)

  
    
  
  Usage Example


defmodule MyApp.ProductListingView do
  use Livex.LivexView
  
  state :selected_category, :string, url?: true
  state :sort_by, :atom, url?: true
  state :modal_open, :boolean
  state :edit_form, MyApp.Components.EditForm
  
  def pre_render(socket) do
    {:noreply,
     socket
     |> assign_new(:modal_open, fn -> false end)
     |> assign_new(:page_title, fn -> "Product Catalog" end)
     |> assign_new(:selected_category, fn -> "all" end)
     |> assign_new(:sort_by, fn -> :name_asc end)
     |> stream_new(:products, [:selected_category, :sort_by], fn assigns ->
       Products.list_available_products(
         category: assigns.selected_category,
         order_by: assigns.sort_by
       )
     end)}
  end
  
  def render(assigns) do
    ~H"""
    <div>
      <h1>{@page_title}</h1>
      
      <.live_component
        :if={@edit_form}
        module={MyApp.Components.EditForm}
        id="edit-form"
        {@edit_form}
        phx-close="close_form"
      />
      
      <!-- Product listing and filters -->
    </div>
    """
  end
  
  def handle_event("close_form", _, socket) do
    {:noreply, assign(socket, :edit_form, nil)}
  end
end

      


      
        Summary


  
    Functions
  


    
      
        assign_from_uri(module, params, uri, socket)

      


    


    
      
        on_mount(module, params, session, socket)

      


    





      


      
        Functions

        


  
    
      
    
    
      assign_from_uri(module, params, uri, socket)



    

  


  


  



  
    
      
    
    
      on_mount(module, params, session, socket)



    

  


  


  


        

      


  

    
Livex.ParamsMapper 
    



      
Handles parameter mapping and type casting for Livex components and views.
This module is responsible for converting parameters between different formats
and ensuring proper type casting based on component attribute definitions.

      


      
        Summary


  
    Functions
  


    
      
        cast(val, arg2)

      


        Cast a value to the specified type.



    


    
      
        decode_json(val, type_check)

      


        Decode a JSON string and verify its type.



    


    
      
        map_params(component, params, id \\ nil)

      


        Map and cast parameters for a given component, optionally scoped by an id.



    


    
      
        parse_bool(val)

      


        Parse a string into a boolean value.



    


    
      
        parse_iso(fun, val)

      


        Parse an ISO formatted string using the provided function.



    


    
      
        safe_atom(val)

      


        Safely convert a string to an existing atom.



    


    
      
        safe_parse(fun, val)

      


        Safely parse a value using the provided function.



    





      


      
        Functions

        


  
    
      
    
    
      cast(val, arg2)



    

  


  

      

          @spec cast(any(), atom()) :: any()


      


Cast a value to the specified type.

  



  
    
      
    
    
      decode_json(val, type_check)



    

  


  

Decode a JSON string and verify its type.

  



    

  
    
      
    
    
      map_params(component, params, id \\ nil)



    

  


  

      

          @spec map_params(module(), map() | nil, any() | nil) :: map()


      


Map and cast parameters for a given component, optionally scoped by an id.

  



  
    
      
    
    
      parse_bool(val)



    

  


  

Parse a string into a boolean value.

  



  
    
      
    
    
      parse_iso(fun, val)



    

  


  

Parse an ISO formatted string using the provided function.

  



  
    
      
    
    
      safe_atom(val)



    

  


  

Safely convert a string to an existing atom.

  



  
    
      
    
    
      safe_parse(fun, val)



    

  


  

Safely parse a value using the provided function.

  


        

      


  

    
Livex.RenderedManipulator 
    



      
A utility library for manipulating Phoenix.LiveView.Rendered structs.

      


      
        Summary


  
    Functions
  


    
      
        attribute_snippets(attributes, assigns)

      


    


    
      
        format_attribute(name, value, prefix \\ nil, type \\ nil)

      


        Formats an attribute for inclusion in an HTML tag.



    


    
      
        manipulate_rendered(mode, rendered, attributes, assigns, options \\ nil)

      


        Manipulates a Phoenix.LiveView.Rendered struct by either injecting attributes or wrapping content.



    


    
      
        maybe_add_dom_id(id)

      


        Conditionally adds an ID attribute to an HTML element.



    


    
      
        maybe_prepend_route(arg1, attributes)

      


    


    
      
        wrap_in_div(rendered, module, assigns)

      


    





      


      
        Functions

        


  
    
      
    
    
      attribute_snippets(attributes, assigns)



    

  


  


  



    

    

  
    
      
    
    
      format_attribute(name, value, prefix \\ nil, type \\ nil)



    

  


  

Formats an attribute for inclusion in an HTML tag.

  
    
  
  Parameters


	name - The attribute name
	value - The attribute value
	prefix - Optional prefix for the attribute name
	type - Optional type for the attribute name


  
    
  
  Returns


	A formatted attribute string


  



    

  
    
      
    
    
      manipulate_rendered(mode, rendered, attributes, assigns, options \\ nil)



    

  


  

Manipulates a Phoenix.LiveView.Rendered struct by either injecting attributes or wrapping content.

  
    
  
  Options


	:inject - Injects attributes into the first static chunk
	:wrap - Wraps the content with a new tag that has the specified attributes


  
    
  
  Parameters


	mode - Either :inject or :wrap
	rendered - The Phoenix.LiveView.Rendered struct to manipulate
	attributes - List of attribute strings to add
	options - Additional options depending on the mode:	For :inject mode:	hook_name - Optional hook name to add


	For :wrap mode:	tag - The HTML tag to wrap content with
	hook_name - Optional hook name to add
	id - Optional ID to add to the wrapping tag






  



  
    
      
    
    
      maybe_add_dom_id(id)



    

  


  

Conditionally adds an ID attribute to an HTML element.
Returns an ID attribute string if the ID is provided, otherwise returns an empty string.

  
    
  
  Examples


iex> maybe_add_dom_id("my-element")
" id=\"my-element\""

iex> maybe_add_dom_id(nil)
""

  



  
    
      
    
    
      maybe_prepend_route(arg1, attributes)



    

  


  


  



  
    
      
    
    
      wrap_in_div(rendered, module, assigns)



    

  


  


  


        

      


  

    
Livex.Utils 
    



      
Utility functions for Livex components and views.
This module provides helper functions for event handling, JavaScript execution,
and state management with dependency tracking.

      


      
        Summary


  
    Functions
  


    
      
        assign_new(socket, key, deps, fun)

      


        Assigns a new value to the socket if it doesn't exist or if any of its dependencies have changed.



    


    
      
        push_emit(socket, event, opts \\ [])

      


        Emits a client-side event from the server.



    


    
      
        push_js(socket, js)

      


        Pushes a JavaScript command to be executed on the client.



    


    
      
        send_message(module, socket, event, payload)

      


    


    
      
        stream_new(socket, key, deps, fun)

      


    





      


      
        Functions

        


  
    
      
    
    
      assign_new(socket, key, deps, fun)



    

  


  

Assigns a new value to the socket if it doesn't exist or if any of its dependencies have changed.
This function is similar to Phoenix.Component.assign_new/3, but with dependency tracking.
It will recompute the value if any of the dependencies have changed in the socket's assigns.

  
    
  
  Parameters


	socket - The LiveView socket
	key - The key to assign the value to
	deps - A list of keys that this value depends on
	fun - A function that returns the value to assign


  
    
  
  Examples


def pre_render(socket) do
  {:noreply,
   assign_new(socket, :location, [:location_id], fn assigns -> 
     MyApp.Domain.get_location!(assigns.location_id)
   end)}
end

  



    

  
    
      
    
    
      push_emit(socket, event, opts \\ [])



    

  


  

Emits a client-side event from the server.
This function allows components to emit events that can be handled by parent components
or views, similar to how JSX.emit works but from server-side code.

  
    
  
  Parameters


	socket - The LiveView socket
	event - The event suffix (will be looked up in assigns as "phx-#{event}")
	opts - Options to pass to the event (default: [])


  
    
  
  Examples


def handle_event("close", _, socket) do
  {:noreply, socket |> push_emit(:close)}
end

  



  
    
      
    
    
      push_js(socket, js)



    

  


  

Pushes a JavaScript command to be executed on the client.

  
    
  
  Parameters


	socket - The LiveView socket
	js - The Phoenix.LiveView.JS struct containing operations to execute


  
    
  
  Returns


	The updated socket


  



  
    
      
    
    
      send_message(module, socket, event, payload)



    

  


  


  



  
    
      
    
    
      stream_new(socket, key, deps, fun)

